

カセットベースのマークカードリーダー用
(M-300) BASICプログラムの開発
— アンケート集計 —

倉島 敬治*・藤原 昇*・高野 豊四郎**

昭和58年後半に附属中学校はNEC PC 8801と周辺システムを導入した。全校で唯一のシステムなので汎用性が要求され、多方面の利用範囲に対応するソフト(プログラム)の準備、開発が必要となった。機器構成は次の通りである。

- 1° パーソナルコンピュータ NEC PC-8801
- 2° ディスプレイ(4000文字) NEC PC-8853
- 3° ドットプリンター(漢字) NEC PC-8822
- 4° マークカードリーダー マミヤ M-300
- 5° データコーダー NEC PC-6081

各1基ずつによる構成で、職員室の一隅に設置されている。

現在、既にパソコンの応用、利用範囲はかなり開発が進み、広がりつつあるが、一応、領域別にプログラムを用意した。

教育工学センターのパソコンがNECのPC8001シリーズで構成されているので、互換性を考慮した上での、パソコンシステムの導入であったこともあって、当センターも当面稼働するプログラムの供給と開発のサポートをすることになった。

各領域少なくとも1プログラム以上、供給する支援体制をとることにした。

- 1) 教科教育 (1) 提示用 a 数学 済み
b 理科 済み
c 社会 済み
(2) CAT a 未定 済み
ドリル b 数学 済み
(3) 生徒自身の開発・教育用
a 未定
- 2) 学習分析・指導 (1) 学習分析 S-P表
CMI I式P-P分析
(2) 成績処理 一覧表作成
- 3) 教務関係 (1) 時間割作成 開発済み
(2) 教室の割あて 未定
(3) その他 未定
- 4) 生活指導関係 (1) アンケート処理 試作
本プログラム開発
(2) その他 未定
- 5) 事務・図書資料 (1) 文書の発行 ワープロ
同窓会名簿他 (2) 名簿の管理 ワープロ

* 岩手大学教育学部附属教育工学センター

** 岩手大学教育学部附属中学校

(3) その他 未定

拡がりも深さも増して来ているパソコン応用の範囲に総て対応することは、開発能力の限界もあって不可能であるが、少なくとも上記の5領域には1つ以上のソフト(プログラム)を供給し、まず利用することから始め、現場の要求に対応したプログラムに改良・改善し、更に余裕があれば、新開発へ協力的体制を組むことにした。

プログラム開発の目的

前記パーソナルコンピュータシステムの機器構成で明らかなように、マークカードリーダー(M-300)が導入されているが、これはN・DISK BASICで稼働するよう別売ソフト(FD)がサポートされている。しかし、システムには予算の関係でFDは未購入になっており、アンケート集計、S-P表作成はマークカードでは不可能である。

テンキーを使ってデータ入力するプログラムを開発すれば問題は解決するわけだが、入力の労力、時間が現場教師に負担となり、マークのカード方式を採用した意味がなくなる。

購入した機器(ハード)があるのにそれを稼働させるプログラムがなく、FD設置まで放置するのは困るので、カセット(CMT)ベースで動くマークカードリーダー用のプログラムをなんとか開発して欲しいという、現場の要求に答えることになったのが第一目的である。

CMTはFDと比べ、容量、書き込み読み出し速度は遅く、正確さも劣り、フリーアクセスもきかない為、当然処理出来るデータ数も制限されてしまう。その上、総てメモリ上で処理する為、CPUのメモリの制限内で最大限その機能を発揮させる必要がある。機械語で思い切った節約、合理化したプロの作品はSpeedy μ で実察に動いているが、それでも500人で約25項目の処理しか出来ない。(24KB)

とりあえず、当方の開発力も考えてN・BASIC+マシン語でCMTベースのプログラムを開発し、学年(4クラス)単位のアンケート処理を可能にし、FD導入時点でプログラムを変更・修正して処理対象数の増加をはかることにした。

既存のFD用プログラムを検討したところ、出力フォーマットが実際の利用者の要求に合わないことが判明し、要求を出して貰った。

- 1) 回答者番号順の入力原表がほしい。
- 2) 単純集計表の実数、パーセント表示の集計表がほしい。
- 3) 操作は画面表示で、全くパソコンを扱ったことのない人でも容易に動かせるものがほしい。以上のような要望にまとめられていた。

プログラム開発における問題解決・アイデア

- 1) マミヤカードリーダーからのデータ転送モードは大別すると2モードがある。
 - (1) カードにマークされた数字をそのままアスキーコードにして1欄に対し1バイトのデータ転送する。
 - (2) 1欄を上中下に3分割し、それらひとつひとつについて16進数の文字におきかえてアスキーコードに転送する。一見、数値データの転送には(1)の方が有利に見えるが、空欄があるとエラーと見なしてしまうので、コメントカードなどの読み取りも出来るように、(2)のモードを選択した。
- 2) マミヤカードリーダーのサンプルプログラムでは1枚につきデータエリア240バイトを使用しているが、FDを使用するので可能だが、CMTで200名の処理をするにはメモリエリアだけで48Kバイト必要となり、実行不可能なので1枚分のデータを読む毎にデータ変換を行いBASICに文字列として渡すことにした。
- 3) データ変換ルーチンの作成
第1段階：BASICを使いデータ変換を実行し

たら、約30秒/枚もかかり実用的にならなかった。

第2段階：機械語を使い、機械語エリア内に変換データをスタックし、BASICからPEEK命令で読み込むことに改良した。1秒/枚でデータ変換が出来るようになったが、なお遅い感じがした。

第3段階：BASICのVARPTR命令を使い変換データを直接BASICの文字変数のスタックエリアに移すようにしたところ、約0.5秒/枚となり、スピード感、実用性に耐える格段の向上がみられた。

番号	1000～2150	メインプログラム
	1000～1180	初期設定
	1300～	カード入力指定
	1500～	マークカード読み込み
	1700～	エラー処理
	1800～	コメント出力
	2000～	データ処理
	3000～	メニュー画面表示
	4000～	入力原表
	5000～	集計表出力 実数, %表
	6000～	クロス集計表
	8000～	機械語DATA

要望の入力原表（生徒番号順）と実数及び%表示の単純集計表のプログラムは5000～に入っている。


2) 特徴は、SPEEDY μ (NEC) 用のデータカードを処理することが出来る。両方のシステムで処理が可能である。

3) SPEEDYのアンケート集計プログラムはプロテクトされていて、出力フォーマットを変えることが出来ないが、本プログラムでは変更、追加が一寸工夫すれば可能となる。

5000～を変更すれば、成績処理、S-P表作成などの処理をマークカードデータ入力で可能となる。

プログラム実行方法（使用方法）

- 1) システム各機器の電源を入れる。
- 2) キーボードよりnew on 1 RETURN と入力し、N・BASICモードにする。
- 3) プログラムテープをデータコーダに入れ巻き戻しておく。
- 4) キーボードからcloud "アンケート" RETURN と入力しデータレコーダのplayボタンを押す。
- 5) データレコーダが止まり、画面にOKと表示されたら f・5 キーによりプログラムを実行する。
- 6) 画面（第2図）を見ながら、選択肢数を決める。（5肢のとき・・・1, 10肢のとき・・・2）



第1図 アンケート集計プログラムメモリマップ

以上がハード含みのプログラム改良点である。

プログラムの構成

1) 文末第1表にプログラムリストを掲示しておいたが、プログラムの構成は次の通りである。

出力：各45秒

全処理：約7～8分

4) 今後の課題

(1) 機械語部分を1つにまとめ、データ転送と同時にデータ変換を行うようにすれば、原データメモリは $240 - 80 = 160$ バイト節約出来、スピードアップになる。

(2) BASICで行っているデータ集計部分も機械語に変換すれば、処理時間はより短縮する。

(3) プログラム行番号2000～2150のデータ集計部分、3000～6390の出力部分の変更によって、アンケート集計以外の処理を行うことも可能である。

現在、パソコンの普及で量販により周辺器のコストダウンは時日進んでいる。フロッピーディスクも既に10万円台を切り始めているので、FDベ

ースによるプログラム、データのやりとりが速く、正確に、容量の増大とともに実現するだろう。

教育工学センターのスタッフとしてはCMTベースのマークカードリーダー用のプログラム開発という課題を投げかけられたので、課題解決のため教育工学の立場から、実用的レベルでなんとか答えることが出来たのではないかと考えている。

参考文献

1) 倉島敬治(1982): マークカード入力処理をベースにしたローコストマイコンシステムの構成, 岩手大学教育工学センター 教育工学研究第4号, 67～72。

2) 電波新聞社(1983): Z80マイコンプログラムテクニック

3) マミヤ光機(株)(1983): M-300取扱説明書

第1表

```
1000 REM ■■■ MAIN ■■■
1010 CLEAR 13000, &HE5FF:DEFINT A-Z:DEFUSR0=&HE600:DEFUSR1=&HE690
1020 DIM B$(200), MS(50)
1030 WIDTH80, 25:CONSOLE0, 25, 0, 1:PRINT CHR$(12)
1040 COLOR6:LINE(15, 5)-(58, 10), "*", B:COLOR7
1050 LOCATE25, 7:PRINT"アンケート シュウケイ プログラム"
1060 RESTORE 8020
1070 FOR AD=&HE600 TO &HE7DF:READ D$:DT=VAL("&H"+D$):POKE AD, DT:NEXT AD
1100 D$=STRING$(80, 32):AD=VARPTR(D$)
1110 D=PEEK(AD+1):POKE &HE697, D:D=PEEK(AD+2):POKE &HE698, D
1120 LOCATE20, 15:PRINT"アンケートカード" ノ センタクシハ ナンコ テ"スカ?"
1130 LOCATE25, 17:PRINT"(1)=5コ マタハ (2)=10コ";:A$=INPUT$(1)
1140 IF (A$<"1")*(A$<"2") THEN 1130
1150 A=VAL(A$):A2=A*5+1:DIM SE$(A2), T(A2)
1160 FOR I=1 TO A*5
1170 IF I=10 THEN SE$(I)="0" ELSE SE$(I)=CHR$(I+48)
1180 NEXT I:SE$(A2)="B"
1300 REM === カート" ニュウリョク シテイ ===
1310 PRINT CHR$(12):COLOR1:LINE(10, 15)-(40, 20), "■", BF:COLOR2
1320 LINE(8, 12)-(38, 17), "■", BF:COLOR5:LINE(6, 9)-(36, 14), "■", BF
1330 LOCATE5, 14:PRINT"▼":LOCATE36, 8:PRINT"▲"
1340 COLOR6:LINE(5, 8)-(35, 13), "■", BF:COLOR4:LINE(3, 5)-(33, 10), "■", BF
1350 LOCATE3, 5:PRINT"▲":COLOR7
1360 LOCATE13, 8:PRINT"END カート":LOCATE15, 12:PRINT"カイトウ カート"
1370 LOCATE15, 16:PRINT"モンク"イ シテイ カート":LOCATE20, 19:PRINT"コメント カート"
1380 LOCATE45, 10:PRINT"ヒタ"リ ノ ヨウニ カート" ラ ソロエデ"
1390 LOCATE47, 12:PRINT"リーダ"-ニ セット シテクダ"サイ!!"
1400 LOCATE45, 14:PRINT"OK ナラ (SPACE) キーヲ オシテクダ"サイ.";
1410 A$=INPUT$(1):IF A$<" " THEN 1410
1500 REM === マークカード" ヨミコミ ===
1510 PRINT CHR$(12):LOCATE30, 11:PRINT"カート"ヲ ヨミコミマス"
```

```

1520 AA=USRO(0):ST=PEEK(&HE8F0):IF ST<>48 THEN 1700
1530 AA=USR1(0):CM$=MID$(D$,11,30)
1540 AA=USRO(0):ST=PEEK(&HE8F0):IF ST<>48 THEN 1700
1550 AA=USR1(0):MO$=MID$(D$,11,50):IO=1
1560 AA=USRO(0):ST=PEEK(&HE8F0):IF ST<>48 THEN 1700
1570 AA=USR1(0):B$(IO)=LEFT$(D$,60)
1580 IF MID$(B$(IO),3,3)<>"000" THEN IO=IO+1:GOTO 1560
1590 N=IO-1:IO=1:FOR I=1 TO 50
1600 IF MID$(MO$,I,1)="-" THEN MS(I)=1:IO=IO+1 ELSE MS(I)=0
1610 NEXT I:M=IO-1:DIM MO(M),P(M,A2)
1620 GOTO 1800
1700 REM === エラー ショリ ===
1710 COLOR2:LOCATE28,15:PRINT"カート" ヨミトリ エラー !!":COLOR7
1720 LOCATE30,17:PRINT"カート" リート エラー
1730 LOCATE25,19:PRINT"モウイチト" ヤリナオシテ クタ"サイ !!
1740 FOR I=1 TO 5000:NEXT I:GOTO 1300
1800 REM === コメント シュツリョク ===
1810 LPRINT STRING$(80,"*"):LPRINT:LPRINT
1820 LPRINT"----- アンケート シュウケイ -----":LPRINT:LPRINT
1830 LPRINT TAB(5);CM$
1840 LPRINT:LPRINT" ニンス"ウ =" ;N:LPRINT" モンタ"イ =" ;M:LPRINT
2000 REM === DATA ショリ ===
2010 PRINT CHR$(12):LOCATE30,11:PRINT"シハ"ラク オマチクタ"サイ !!
2020 FOR I=1 TO N-1:FOR J=I+1 TO N
2030 IF VAL(MID$(B$(J),8,3))<=VAL(MID$(B$(I),8,3)) THEN SWAP B$(I),B$(J)
2040 NEXT J,I
2050 FOR I=1 TO N:B0$=LEFT$(B$(I),10):J0=1:FOR J=1 TO 50
2060 IF MS(J)=0 THEN 2120
2070 D$=MID$(B$(I),10+J,1):D=VAL(D$)
2075 IF D$<"0" OR D$>"9" THEN 2095
2080 IF A=1 THEN IF DMOD2=1 THEN 2095 ELSE D=(D+2)/2:GOTO 2090
2085 D=D+1:IF D=10 THEN D$="0":GOTO 2100
2090 D$=CHR$(48+D):GOTO 2100
2095 D=A2:D$="B"
2100 B0$=B0$+D$:P(J0,D)=P(J0,D)+1:J0=J0+1
2120 NEXT J:B$(I)=B0$:NEXT I
2130 IO=1:FOR I=1 TO 50
2140 IF MS(I)=1 THEN MO(IO)=I:IO=IO+1
2150 NEXT I
3000 REM ■■■ フ'リント アウト ■■■
3010 PRINT CHR$(12):LOCATE20,4:PRINT"フ'リント ニ シュツリョク シマス."
3020 LOCATE22,6:PRINT"シュツリョク シテイヲ シテクタ"サイ!!"
3030 LOCATE24,8:PRINT"(1)... ニュウリョク ケンビ"ョウ"
3040 LOCATE24,10:PRINT"(2)... タンシ"エン シュウケイ "
3050 LOCATE24,12:PRINT"(3)... ハ"セント シュウケイ "
3060 LOCATE24,14:PRINT"(4)... クロス シュウケイ "
3070 LOCATE24,16:PRINT"(0)... END ";
3080 A$=INPUT$(1):IF A$<"0" OR A$>"4" THEN 3080
3090 OP=VAL(A$):IF OP=0 THEN 3120
3100 ON OP GOSUB 4000,5000,5000,6000
3110 GOTO 3000
3120 LPRINT:LPRINT:LPRINT STRING$(80,"*"):LPRINT:LPRINT
3130 PRINT CHR$(12):LOCATE35,11:PRINT"コ"クロウ サマテ"シタ !!
3140 END
4000 REM ◆◆◆ ニュウリョク ケンビ"ョウ ◆◆◆
4010 PRINT CHR$(12):LOCATE30,11:PRINT"フ'リント シマス !!
4020 LPRINT TAB(15);"***** ニュウリョク ケンビ"ョウ *****":LPRINT
4030 LPRINT" モンタ"イ ハ"ンコ"ウ":LPRINT TAB(5);
4040 M1=1:IF M<35 THEN M2=M:M3=0 ELSE M2=30:M3=1
4050 FOR I=M1 TO M2:MM=INT(MO(I)/10)
4060 IF MM=0 THEN LPRINT" "; ELSE LPRINT USING"##";MM;
4070 NEXT I:LPRINT:LPRINT TAB(5);
4080 FOR I=M1 TO M2:MO=MO(I)-INT(MO(I)/10)*10:LPRINT USING"##";MO;:NEXT I
4090 LPRINT:LPRINT" セイト"

```

```

4100 FOR I=1 TO N:LPRINT USING"### ";VAL(MID$(B$(I),8,3));:FOR J=M1 TO M2
4110 LPRINT MID$(B$(I),10+J,1);" ";
4120 NEXT J:LPRINT:NEXT I:LPRINT:LPRINT
4130 IF M3=1 THEN M1=31:M2=M:M3=0:GOTO 4050
4140 RETURN
5000 REM ◆◆◆ シュウケイ ヒョウ ◆◆◆
5010 PRINT CHR$(12):LOCATE30,11:PRINT"フ'リント シマス !!"
5020 LPRINT TAB(15);
5030 IF OP=2 THEN LPRINT"***** タンシ'ユン シュウケイ ヒョウ *****":GOTO 5050
5040 LPRINT"***** シュウケイ ヒョウ (%) *****"
5050 LPRINT:LPRINT TAB(20);"センタクシ ハ'ンゴ'ウ":LPRINT TAB(11);
5060 FOR I=1 TO A2:LPRINT" ";SE$(I);:NEXT I:LPRINT:LPRINT"モンタ'イ No. r";
5070 FOR I=1 TO A2:LPRINT"-----";:NEXT I
5080 IF OP=2 THEN LPRINT"- " ELSE LPRINT"-%"
5090 FOR I=1 TO M:LPRINT USING"###I";MO(I);
5100 FOR J=1 TO A2:P=P(I,J)
5110 IF OP=2 THEN LPRINT USING"###";P;:GOTO 5130
5120 LPRINT USING"###";INT(P*100/N+.5);
5130 NEXT J:LPRINT:NEXT I:LPRINT TAB(10);"L";
5140 FOR I=1 TO A2:LPRINT"-----";:NEXT I:LPRINT"-":LPRINT:LPRINT
5150 RETURN
6000 REM ◆◆◆ クロス シュウケイ ヒョウ ◆◆◆
6010 LPRINT TAB(15);"***** クロス シュウケイ ヒョウ *****":LPRINT
6020 FOR I=1 TO A2:T(I)=0:NEXT I
6030 PRINT CHR$(12):LOCATE25,8:PRINT"クロス シュウケイ ラ シマス."
6040 LOCATE20,10:INPUT"1ツメ ノ モンタ'イ No. ラ ニュウリョク シテク'サイ! ";A$:M1=VAL(A$)
6050 FOR I=1 TO M:IF MO(I)=M1 THEN M3=I ELSE NEXT I:GOTO 6040
6060 LOCATE20,12:INPUT"2ツメ ノ モンタ'イ No. ラ ニュウリョク シテク'サイ! ";A$:M2=VAL(A$)
6070 FOR I=1 TO M:IF MO(I)=M2 THEN M4=I ELSE NEXT I:GOTO 6060
6080 IF M1=M2 THEN 6040
6090 LOCATE25,15:PRINT"シュツリョク ノ カクチハ ?"
6100 LOCATE28,17:PRINT"(0)=シ'ッスウ マクハ (1)=%";
6110 A$=INPUT$(1):IF A$<>"0" AND A$<>"1" THEN 6110
6120 S0=VAL(A$):LOCATE30,20:COLOR3:PRINT"フ'リント シマス !!":COLOR7
6130 LPRINT TAB(16);"No. ";:LPRINT USING"###";MO(M4):LPRINT
6140 LPRINT TAB(11);:FOR I=1 TO A2:LPRINT" ";SE$(I);:NEXT I
6150 IF S0=0 THEN LPRINT" TOTAL" ELSE LPRINT" TOTAL (%)"
6160 LPRINT TAB(10);"r";:FOR I=1 TO A2:LPRINT"-----";:NEXT I:LPRINT"ー"
6170 FOR I=1 TO A2:T=0
6180 IF I=A*2 THEN LPRINT"No. ";:LPRINT USING"###";MO(M3);
6190 LPRINT TAB(9);SE$(I);": ";
6200 FOR J=1 TO A2:Z=0:FOR K=1 TO N
6210 M1$=MID$(B$(K),M1+10,1):M2$=MID$(B$(K),M2+10,1)
6215 IF (M1$=SE$(I))*(M2$=SE$(J)) THEN Z=Z+1:T(J)=T(J)+1
6220 NEXT K:T=T+Z
6230 IF S0=1 THEN Z=Z*100/N+.5
6240 LPRINT USING"###";Z;
6250 NEXT J:LPRINT" : ";
6260 IF S0=1 THEN T=T*100/N+.5
6270 LPRINT USING"###";T
6280 NEXT I
6290 LPRINT TAB(10);"L";:FOR I=1 TO A2:LPRINT"-----";:NEXT I:LPRINT"ー"
6300 LPRINT TAB(5);"TOTAL:";:FOR I=1 TO A2
6310 IF S0=1 THEN T(I)=T(I)*100/N+.5
6320 LPRINT USING"###";T(I);
6330 NEXT I:LPRINT" : ";
6340 IF S0=0 THEN LPRINT USING"###";N ELSE LPRINT" 100"
6350 LPRINT:LPRINT
6360 LOCATE25,20:PRINT"クロス シュウケイ ラ ツツ'ケマスカ ? (y/n) ";
6370 A$=INPUT$(1):IF (A$="Y")+(A$="y") THEN 6020
6380 IF (A$<>"N")*(A$<>"n") THEN 6370
6390 RETURN
8000 REM ■■■ マシンゴ` Data ■■■
8010 REM --- カート` ヨミトリ ---

```

```

8020 DATA CD, 10, E6, CD, 1B, E6, 3A, F0, E8, FE, 30, C0, CD, 2C, E6, C9
8030 DATA 3E, 11, CD, 54, E6, 3E, 0D, CD, 54, E6, C9, 3E, 05, CD, 54, E6
8040 DATA 3E, 0D, CD, 54, E6, CD, 6B, E6, 32, F0, E8, C9, 3E, 10, CD, 54
8050 DATA E6, 3E, 43, CD, 54, E6, 3E, 0D, CD, 54, E6, 3E, 00, 32, F1, E8
8060 DATA 21, 00, E8, CD, 6B, E6, 77, 23, 3A, F1, E8, 3C, 32, F1, E8, FE
8070 DATA F0, 38, F0, C9, D3, B1, 3E, 01, D3, B3, CD, 85, E6, 3E, 00, D3
8080 DATA B3, CD, 85, E6, DB, B2, CB, 47, 20, FA, C9, 3E, 02, D3, B3, CD
8090 DATA 85, E6, 3E, 00, D3, B3, CD, 85, E6, DB, B2, CB, 4F, 20, FA, CD
8100 DATA 85, E6, DB, B0, C9, 3E, 0A, 3D, 20, FD, C9, 00, 00, 00, 00, 00
8110 REM --- data ヘンカン ---
8120 DATA 21, 00, E8, 22, F0, E7, 21, 00, E9, 22, F2, E7, 97, 32, F4, E7
8130 DATA 2A, F0, E7, 7E, 32, F6, E7, 23, 7E, 32, F7, E7, 23, 7E, 32, F8
8140 DATA E7, FE, 46, 28, 2F, 01, 30, 30, 2A, F6, E7, AF, ED, 42, 20, 05
8150 DATA CD, 00, E7, 18, 03, CD, 40, E7, 2A, F0, E7, 23, 23, 23, 22, F0
8160 DATA E7, 2A, F2, E7, 23, 22, F2, E7, 3A, F4, E7, 3C, 32, F4, E7, FE
8170 DATA 50, 20, BD, C9, 2A, F2, E7, 36, 20, 23, 22, F2, E7, 3A, F4, E7
8180 DATA 3C, 32, F4, E7, FE, 50, 20, EC, C9, 00, 00, 00, 00, 00, 00, 00
8190 DATA 2A, F2, E7, 06, 31, 36, 26, CD, D0, E7, C8, 06, 32, 36, 2D, CD
8200 DATA D0, E7, C8, 06, 34, 36, 30, CD, D0, E7, C8, 06, 38, 36, 31, CD
8210 DATA D0, E7, C8, 06, 39, 36, 41, CD, D0, E7, C8, 06, 41, 36, 4A, CD
8220 DATA D0, E7, C8, 06, 43, 36, 2F, CD, D0, E7, C8, 36, 20, C9, 00, 00
8230 DATA 2A, F2, E7, 06, 30, 36, 31, CD, D0, E7, 28, 1E, 06, 31, 36, 41
8240 DATA CD, D0, E7, 28, 15, 06, 32, 36, 4A, CD, D0, E7, 28, 0C, 06, 34
8250 DATA 36, 52, CD, D0, E7, 28, 03, 36, 20, C9, 01, 30, 31, 3E, 01, CD
8260 DATA D5, E7, 28, 4C, 01, 30, 32, 3E, 02, CD, D5, E7, 28, 42, 01, 30
8270 DATA 34, 3E, 03, CD, D5, E7, 28, 38, 01, 30, 38, 3E, 04, CD, D5, E7
8280 DATA 28, 2E, 01, 31, 30, 3E, 05, CD, D5, E7, 28, 24, 01, 32, 30, 3E
8290 DATA 06, CD, D5, E7, 28, 1A, 01, 34, 30, 3E, 07, CD, D5, E7, 28, 10
8300 DATA 01, 38, 30, 3E, 08, CD, D5, E7, 28, 06, 2A, F2, E7, 36, 20, C9
8310 DATA 21, FA, E7, 46, 2A, F2, E7, 7E, 80, 77, C9, 00, 00, 00, 00, 00
8320 DATA 3A, F8, E7, B8, C9, 21, FA, E7, 77, 2A, F6, E7, AF, ED, 42, C9
9000 REM
9010 REM *****
9020 REM **
9030 REM ** PC-8001 アンケート シュウケイ プログラム **
9040 REM ** (マークシート ニュウリョク) **
9050 REM ** 1984.1.31 by N.F Ver. 2.1 **
9060 REM **
9070 REM *****

```

第2表 アンケート入力原表

----- アンケート シュウゲイ -----

/ EXAMPLE DATA /

ニンスウ = 40
モンダイ = 20

***** ニュウヨク ケンビョウ *****

セイト	モンダイ ハンゴウ																			
	1	2	3	4	5	6	7	8	9	0										
1	3	2	2	3	2	4	3	4	3	3	3	2	2	1	3	1	3	4	4	1
2	3	3	2	1	2	2	2	3	1	2	2	2	2	1	3	1	2	1	1	1
3	3	3	2	2	2	2	2	3	2	3	3	2	2	1	3	1	3	1	1	1
4	3	3	2	2	2	2	3	1	2	2	2	2	1	3	1	3	3	3	1	1
5	3	4	2	1	2	2	2	2	1	2	2	3	4	1	1	2	3	2	3	4
6	3	3	2	4	2	1	2	3	1	3	1	2	2	1	3	1	2	2	1	1
7	3	3	2	2	2	2	3	1	2	2	2	2	1	3	1	1	3	3	1	1
8	3	3	2	2	2	2	4	2	3	1	2	2	1	4	1	4	1	3	1	1
9	3	3	2	2	2	1	2	3	1	3	3	1	4	3	3	1	3	1	2	1
10	3	3	2	3	2	2	2	4	2	3	3	2	2	1	3	2	1	1	2	2
11	3	3	2	2	2	2	3	1	3	1	3	1	2	2	1	3	1	4	1	1
12	3	3	2	2	2	2	3	1	2	2	2	2	1	3	3	1	1	4	1	1
13	3	4	2	1	2	3	2	4	1	3	4	3	2	1	3	2	4	2	1	3
14	3	3	2	2	2	2	3	1	2	2	2	2	1	3	1	4	3	3	3	3
15	3	3	2	1	2	2	2	3	1	2	2	2	2	1	3	2	2	3	3	1
16	3	3	2	1	2	2	2	3	1	2	2	2	2	1	3	2	3	3	2	1
17	3	3	2	2	2	4	2	3	3	2	2	2	2	1	3	1	4	3	3	1
18	3	3	2	2	1	2	2	1	3	1	1	2	2	1	3	1	4	1	2	1
19	3	3	2	2	2	2	3	2	2	2	2	2	2	1	3	1	3	1	2	1
20	3	3	2	2	2	2	4	1	3	1	2	2	1	3	2	3	1	3	2	1
21	3	3	2	4	2	4	2	3	2	3	3	2	2	1	3	1	3	1	2	1
22	3	4	2	1	2	2	2	4	1	3	4	3	1	2	1	3	2	4	1	1
23	3	3	2	2	2	2	2	3	3	2	2	2	2	1	1	2	4	1	4	1
24	3	3	2	2	2	2	3	2	2	2	2	2	2	1	3	1	3	1	2	1
26	3	3	2	4	3	2	2	3	3	3	3	1	1	2	3	3	3	3	3	1
27	3	3	2	2	2	2	2	3	2	3	3	2	2	2	3	2	4	1	1	3
28	3	3	2	2	2	2	2	3	2	2	2	2	2	1	1	3	1	1	3	3
29	3	3	4	4	3	2	2	2	2	2	2	2	2	2	3	2	1	1	4	1
30	3	3	2	2	2	2	2	3	1	3	3	2	2	1	3	1	4	3	1	4
31	3	3	2	2	2	2	2	2	2	2	2	2	2	1	3	1	3	1	2	2
32	3	3	3	2	2	2	2	3	1	2	2	2	2	1	3	1	3	3	3	1
33	1	3	4	2	2	2	2	3	1	4	4	3	3	2	3	5	5	1	1	1
34	3	2	5	2	2	3	2	3	2	3	1	2	2	2	3	2	2	1	3	3
35	3	3	2	4	2	2	2	3	B	2	2	2	2	1	3	1	3	1	2	1
36	2	2	2	1	2	2	2	1	3	1	4	2	2	2	3	2	4	5	1	1
38	3	3	2	4	2	2	2	3	1	4	1	2	2	3	3	2	2	1	1	1
39	3	3	2	1	2	2	2	3	B	2	4	1	4	1	3	1	2	1	1	2
41	3	3	2	2	2	2	2	3	B	3	3	3	3	2	3	1	4	1	3	4
42	3	3	2	2	2	2	2	3	1	2	2	2	2	1	3	1	3	3	3	1
43	1	3	2	2	2	2	2	1	1	2	3	2	1	3	1	3	1	1	1	1

第3表

***** タンラジョン シュウゲイ ヒョウ *****

モンダイ No.	セントラ ハンゴウ					B
	1	2	3	4	5	
1	2	1	37	0	0	0
2	0	3	34	3	0	0
3	0	36	1	2	1	0
4	8	24	2	6	0	0
5	1	37	2	0	0	0
6	2	33	2	3	0	0
7	0	39	1	0	0	0
8	2	4	28	6	0	0
9	20	11	6	0	0	3
10	3	19	16	2	0	0
11	7	19	9	5	0	0
12	4	30	6	0	0	0
13	3	32	2	3	0	0
14	29	8	3	0	0	0
15	4	1	34	1	0	0
16	24	11	4	0	1	0
17	4	7	17	11	1	0
18	22	4	11	2	1	0
19	14	9	13	4	0	0
20	28	4	5	3	0	0

第4表

***** シュウゲイ ヒョウ (%) *****

モンダイ No.	セントラ ハンゴウ					%
	1	2	3	4	5	
1	5	3	93	0	0	0
2	0	8	85	8	0	0
3	0	90	3	5	3	0
4	20	60	5	15	0	0
5	3	93	5	0	0	0
6	5	83	5	8	0	0
7	0	98	3	0	0	0
8	5	10	70	15	0	0
9	50	28	15	0	0	8
10	8	48	40	5	0	0
11	18	48	23	13	0	0
12	10	75	15	0	0	0
13	8	80	5	8	0	0
14	73	20	8	0	0	0
15	10	3	85	3	0	0
16	60	28	10	0	3	0
17	10	18	43	28	3	0
18	55	10	28	5	3	0
19	35	23	33	10	0	0
20	70	10	13	8	0	0

第5表

***** クロス ショウゲイ ヒョウ *****

No.17

		1	2	3	4	5	B	TOTAL
No. 3	1:	0	0	0	0	0	0	0
	2:	3	6	16	11	0	0	36
	3:	0	0	1	0	0	0	1
	4:	1	0	0	0	1	0	2
	5:	0	1	0	0	0	0	1
	B:	0	0	0	0	0	0	0
TOTAL:		4	7	17	11	1	0	40

No.17

		1	2	3	4	5	B	TOTAL (%)
No. 3	1:	0	0	0	0	0	0	0
	2:	8	15	40	28	0	0	90
	3:	0	0	3	0	0	0	3
	4:	3	0	0	0	3	0	5
	5:	0	3	0	0	0	0	3
	B:	0	0	0	0	0	0	0
TOTAL:		10	18	43	28	3	0	100